

The sound chip - Introduction.

@.1

A general description of the SN76489A sound chip has already been provided in appendix 7 in the first half of this manual.

This section is concerned primarily with using that information in conjunction with the MTX assembler.

An important point to note is that the MTX has no internal speaker.

Sound output is directed through the television speaker, or as an option through the standard Hi Fi socket at the rear right of the MTX.

Sound chip write only registers.

@.2

Referring to Table 4, appendix 7 in the first half of this manual, it can be seen that the sound chip has eight write only registers.

Each of these registers perform different functions as listed in the first half of this manual. (Page 242 - Table 4).

An important point to note is that the sound chip requires a minimum transfer delay time of 32 microseconds between each data byte output to it.

Sound I/O interface routines.

@.3

All sound chip manipulation can be performed with ease using the following routines :-

- 1) SOFF - Switch off all existing sound.
- 2) SETVOL - Set sound channel volume.
- 3) SETFRE - Set sound channel frequency.

The routines are listed below and are fully documented.

The sound routines only require the use of the C register and DE register pair, which are unaffected after calling any sound routine.

Minimum required timing delays of 32 microseconds are built

into all of the sound routines.

```
;  
; CORE SOUND I/O SECTION  
;  
; PLEASE SEE ACCOMPANYING TEXT FOR FULLER DESCRIPTION OF HOW THESE  
; ROUTINES WORK. (PART 1 OF THIS MANUAL - PAGES 239 TO 244  
; INCLUSIVE)  
;  
; PARAMETERS ARE PASSED TO THE SOUND I/O ROUTINES VIA REGISTER C  
; AND THE DE REGISTER PAIR ONLY.  
; NO OTHER PARAMETERS ARE NEEDED.  
;  
; NO REGISTERS AFFECTED ON EXIT - VRAM READ/WRITE POINTERS  
; UNAFFECTED ON EXIT.  
;  
; UTILITY ROUTINE SOFF REQUIRES NO PARAMETERS AND WILL KILL ALL  
; EXISTING SOUNDS WHEN CALLED.  
;  
; REGISTER C.  
; -----  
;  
; USED TO UPDATE SOUND REGISTERS 1, 3 AND 5, WHICH ARE THE  
; ATTENUATION (VOLUME) CONTROLS FOR PURE SOUND CHANNELS 1, 2 AND  
; 3 RESPECTIVELY. THE BYTE SET UP IS AS BELOW :-  
;  
; CHANNEL      HIGH NIBBLE      LOW NIBBLE  
; -----  
; 1            1                0 TO 15  
; 2            2                0 TO 15  
; 3            3                0 TO 15  
;  
; TO UPDATE NOISE ATTENUATION (VOLUME) CONTROL HELD WITHIN  
; SOUND REGISTER 7, USE THE BYTE SET UP FOR REGISTER C AS BELOW :-  
;  
; HIGH NIBBLE      LOW NIBBLE  
; -----  
; 4                0 TO 15  
;  
; THE FOUR BITS HELD IN THE LOW NIBBLE OF REGISTER C IS THE VOLUME  
; LEVEL UPDATE VALUE.  
;  
; ITS RANGE IS FROM 0 TO 15 INCLUSIVE.  
;  
; 0 IS THE LOWEST LEVEL = VOLUME OFF.  
; 15 IS THE HIGHEST LEVEL = MAXIMUM VOLUME.  
;  
; IN BOTH OF THE ABOVE VOLUME CONTROL OPTIONS BIT 7 OF REGISTER C  
; IS IGNORED  
;  
; TO CARRY OUT A VOLUME LEVEL UPDATE, SET UP REGISTER C AS  
; DESCRIBED ABOVE AND THEN CALL THE ROUTINE SETVOL  
;  
; EG :-      LD      C, #10      ; SELECT CHANNEL 1, VOLUME = OFF  
;           CALL    SETVOL      ; PERFORM VOLUME LEVEL UPDATE
```

```

;           RET                               ;RETURN TO CALLING ROUTINE
;
; REGISTER PAIR DE.
;-----
;
; USED TO UPDATE SOUND REGISTERS 0, 2 AND 4, WHICH ARE 10 BIT
; FREQUENCY CONTROL VALUES FOR CHANNELS 1, 2 AND 3 RESPECTIVELY.
; THE REGISTER PAIR BYTE SET UP IS AS FOLLOWS.
;
;           D BYTE                               E BYTE
; CHANNEL   HIGH NIBBLE  LOW NIBBLE   HIGH NIBBLE  LOW NIBBLE
;-----
;     1           1           (10 BIT VALUE FOR CHANNEL FREQUENCY)
;     2           2           (10 BIT VALUE FOR CHANNEL FREQUENCY)
;     3           3           (10 BIT VALUE FOR CHANNEL FREQUENCY)
;
; IN THE ABOVE OPTION BITS 2 AND 3 OF REGISTER D ARE IGNORED
; THE TEN BIT FREQUENCY VALUE HAS A RANGE OF 0 TO 1023 INCLUSIVE.
; 0 IS THE HIGHEST FREQUENCY VALUE.
; 1023 IS THE LOWEST FREQUENCY VALUE.
;
; THE FREQUENCY IS CALCULATED BY THE FOLLOWING FORMULA :-
;
;           FREQUENCY = (VALUE OF REF CLOCK IN HERTZ = 4,000,000 )
;                       DIVIDED BY ( 32 TIMES 10 BIT FREQUENCY VALUE
;                       ABOVE)
;
; REGISTER PAIR DE HAVE ONE OTHER FUNCTION WHICH IS TO UPDATE THE
; NOISE CONTROL REGISTER 6. THE REGISTER PAIR BYTE SET UP IS AS
; FOLLOWS :-
;
;           D BYTE                               E BYTE
; HIGH NIBBLE  LOW NIBBLE   HIGH NIBBLE  LOW NIBBLE
;-----
;     4           FB= 1 OR 0   NF1= 1 OR 0   NFO= 1 OR 0
;
; FOR A FULLER DESCRIPTION OF FB, NF1 AND NFO PLEASE REFER TO
; APPENDIX 7 IN THE FIRST HALF OF THIS MANUAL.
;
; IN BOTH OF THE OPTIONS ABOVE BIT 7 OF REGISTER D IS IGNORED.
;
; TO PERFORM A FREQUENCY UPDATE, SET UP REGISTER PAIR DE AS
; DESCRIBED ABOVE AND THEN CALL THE ROUTINE SETFRE.
;
; EG :-      LD      DE,£11FF           ;SELECT CHANNEL 1, FREQUENCY
;                                     ;= £1FF (HEX), 511 (DECIMAL)
;           CALL    SETFRE             ;PERFORM FREQUENCY UPDATE
;           RET                                     ;RETURN TO CALLING ROUTINE
;
; TO PERFORM A NOISE CONTROL REGISTER UPDATE SET UP REGISTER PAIR
; DE AS DESCRIBED ABOVE AND THEN CALL THE ROUTINE SETFRE.
;
; EG :-      LD      DE,£4100           ;SELECT NOISE CHANNEL,
;                                     ;WHITE NOISE OPTION,
;                                     ;NF1 = 0, NFO = 0.

```

```

;          CALL      SETFRE          ;PERFROM NOISE CHANNEL UPDATE
;          RET              ;RETURN TO CALLING ROUTINE
;
;ACTUAL SOUND ROUTINES SECTION FOLLOWS.
;
;SOFF      -SWITCH OFF VOLUME ON ALL
;          CHANNELS
;
SOFF:      PUSH      BC              ;Save old BC register pair
          LD        C,£10          ;Select channel 1
          ;and set volume to 0 (Off)
          CALL      SETVOL         ;Send sound control byte to
          ;sound chip
          LD        C,£20          ;Select channel 2
          ;and set volume to 0 (Off)
          CALL      SETVOL         ;Send sound control byte to
          ;sound chip
          LD        C,£30          ;Select channel 3
          ;and set volume to 0 (Off)
          CALL      SETVOL         ;Send sound control byte to
          ;sound chip
          LD        C,£40          ;Select channel 4
          ;and set volume to 0 (Off)
          CALL      SETVOL         ;Send sound control byte to
          ;sound chip
          POP       BC             ;Retreive old BC register pair
          RET              ;Return to calling routine
;
;SETVOL-   UPDATE VOLUME FOR ANY OF THE FOUR CHANNELS
;          (SEE DESCRIPTION AT START OF CODE)
;
;          INPUT PARAMETER IN C ON ENTRY
;
;          NO REGISTERS AFECTED ON EXIT
;
;          ROUTINE USES LOCAL TABLE VOLTAB TO PROVIDE CHIP I/O
;          INTERFACE BYTES
;
VOLTAB:    DB        £00,£10,£30
          DB        £50,£70          ;(See description above)
;
SETVOL:    PUSH      AF            ;Save old Acc and flags
          PUSH      BC            ;Save BC register pair
          PUSH      DE            ;Save DE register pair
          PUSH      HL            ;Save HL register pair
;
          LD        A,C           ;Convert sound volume range
          AND      £0F           ;value of 0 = off - 15 = max
          DEC      A              ;to sound chip internal
          AND      £0F           ;representation
          ;Manual part 1 - Page 240,
          ;table 1
          LD        B,A           ;Save new volume bit set up
          ;in register B
          LD        A,C           ;Retreive original volume

```

```

SRL      A      ;byte from register C and
SRL      A      ;extract the channel number
SRL      A      ;at which the volume update
SRL      A      ;is to occur
LD       HL,VOLTAB ;Then using the table VOLTAB
LD       E,A      ;extract the corresponding
LD       D,O      ;channel control bits used
ADD      HL,DE     ;internally by the sound chip
LD       A,(HL)   ;Add the new internally
ADD      A,B      ;formatted control bit set-ups
OR       £80      ;together and set bit 7
CALL    SNDSUB    ;to form a correct format
                    ;'UPDATE ATTENUATOR BYTE'
                    ;Output sound byte to sound
                    ;chip registers

;
POP      HL      ;Retrieve old HL register pair
POP      DE      ;Retrieve old DE register pair
POP      BC      ;Retrieve old BC register pair
POP      AF      ;Retrieve old Acc and flags
RET      ;Return to calling routine

;
;SETFRE- SET UP CHANNEL FREQUENCY OR UPDATE NOISE CHANNEL
;
; INPUT PARAMETER IN DE ON ENTRY
;
; NO REGISTERS AFFECTED ON EXIT
;
; ROUTINE USES LOCAL TABLE FRETAB TO PROVIDE CHIP I/O
; INTERFACE BYTES
;
FRETAB:  DB      £00,£80,£0A0
         DB      £0C0,£0E0      ;(See description above)
;
SETFRE:  PUSH    AF      ;Save old Acc and flags
         PUSH    BC      ;Save BC register pair
         PUSH    HL      ;Save HL register pair
         PUSH    DE      ;Save DE register pair
;
LD       A,D      ;Select high half of noise/
AND      £70      ;frequency control byte pair
CP       £40      ;Filter out any unwanted bits
                    ;and decide wether update
                    ;noise register is to take
                    ;place - in which case
JP       Z,NOISEC ;goto NOISEC (Noise Select)
                    ;
                    ;else drop through to SETFRO
                    ;and perform frequency update

;
SETFRO:  SRL      A      ;Extract the channel number
         SRL      A      ;at which the frequency update
         SRL      A      ;is to occur
         SRL      A      ;Then using the table FRETAB
         LD       E,A      ;extract the corresponding

```

```

LD      D,0      ;channel control bits used
LD      HL,FRETAB ;internally by the sound chip
ADD     HL,DE    ;and place the result in
LD      B,(HL)  ;register B

;

POP     DE      ;Retreive old frequency
PUSH    DE      ;control byte pair and extract
          ;low 4 bits from 10 bit freq
LD      A,E     ;value
AND     £0F    ;Add the previously saved
ADD     A,B     ;channel select bit set up
          ;held in register B to
          ;produce correctly formatted
          ;'UPDATE FREQUENCY' first byte
          ;(See Manual part 1 - page
          ;242)
CALL    SNDSUB  ;Ouput sound byte to sound
          ;chip registers

;

SETFR1: LD      B,£04 ;Now extract upper 6 bits of
SRL     D      ;10 bit frequency value held
RR      E      ;in register pair DE and
DJNZ   SETFR1 ;place into register A ready
LD      A,E    ;for output
          ;This will be the 'UPDATE
          ;FREQUENCY' second byte
          ;(See Manual part 1 - page
          ;242)
CALL    SNDSUB  ;Ouput sound byte to sound
          ;chip registers
JP      SETFIN  ;goto exit routine section

;
NOISEC: LD      A,D ;Extract FB = 1 or FB = 0
AND     1      ;bit from register D and
SLA    A      ;save in register B
SLA    A
LD      B,A
LD      A,E   ;Extract NF1 and NFO from
AND     £01   ;register E and save in
LD      C,A  ;A and C respectively
LD      A,E
AND     £10
SRL     A
SRL     A
SRL     A
ADD     A,B   ;Add together bit set-up
          ;for FB, NF1 and NFO
ADD     A,C   ;leaving the result in A
OR      £0E0 ;Add sound chip internal
          ;representation for noise
          ;channel select bits to
          ;produce correctly formatted
          ;'UPDATE NOISE SOURCE' control
          ;byte (See Manual part 1 -
          ;page 242)

```

```

CALL SNDSUB ;Output sound byte to sound
;chip registers
;
SETFIN: POP DE ;Retreive old DE register pair
POP HL ;Retreive old HL register pair
POP BC ;Retreive old BC register pair
POP AF ;Retreive old Acc and flags
RET ;Return to calling routine
;
SNDSUB: OUT (£06),A ;Deliver sound byte to sound
;output port
IN A, (£03) ;Strobe sound byte at port 6
;into sound chip registers
RET ;Return to calling routine

```

Generating white noise and periodic noise.

Referring to table 2 in the first half of this manual on page 241, it can be seen that selecting white/periodic noise is a simple matter of setting the FB bit to either 1 or 0 respectively, in the noise control byte, and using a section of code similar to that given below :-

```

1) LD DE,£4000 ;Select channel 4 (Noise
;control) - FB=0 therefore
;periodic noise
CALL SETFRE ;Update noise channel
;register contents
RET ;Return to calling routine

2) LD DE,£4000 ;Select channel 4 (Noise
;control) - FB=1 therefore
;white noise
CALL SETFRE ;Update noise channel
;register contents
RET ;Return to calling routine

```

It can be seen in the above code that the noise control bits NFO and NF1 are both zero.

Referring to table 3 on page 241 in the first half of this manual, it can be seen that NFO and NF1 relate to noise generator frequency control.

In one special case though when NFO and NF1 are both equal to 1, noise output is directed through tone generator channel 3. This will allow you to envelope and modulate noise output as if it were pure sound.